
//***************************************************************
//* MACRO:  PGBORDER.WCM
//* PURPOSE:  Creates a page border from a graphic file.
//***************************************************************
Application (A1; "WordPerfect"; Default; "UK")

OnError(End@)
OnCancel(End@)
DlgDefined:=False

If(?Substructure)
MessageBox(; "Not At Main Edit Screen"; "This macro can be played only from the main
editing screen.  Exit all substructures (headers, text boxes, etc.) and play the macro 
again."; IconStop!)
Quit

EndIf

Path:=?PathGraphics
If(Path="")

Path:=?PathProgram
Call(CheckDir@)

EndIf
If(SubStr(Path; StrLen(Path); 1)<>"\")

Path:=Path+"\"
EndIf

Call(Definitions@)

If (Not(?DocBlank))
PosPageTop()
PosLineBegin()

EndIf

HeightOfBox:=?PaperLength
WidthOfBox:=?PaperWidth

If(WidthOfBox>=HeightOfBox)
Port:=0
Land:=1

Else
Port:=1
Land:=0

EndIf

DlgDefined:=True
DialogDefine("Dlg1"; 50; 50; 172; 103; 1+2+16; "Graphic Page Borders")



DialogAddListBox("Dlg1"; "Lb1"; 8; 8; 75; 75; 1; Choice)
For(Count; 1; Count<=MaxBorder; Count+1)

DialogAddListItem("Dlg1"; "Lb1"; Border[Count; 1])
EndFor
DialogHandle(hwndLb1; "Dlg1"; "Lb1")
DLLLoad(UserLink; "User")

// Select the first item
DLLCall(UserLink; "SendMessage"; nVoid:INTEGER;{LoWord(hwndLb1); LoWord(1031); 
LoWord(0); 0})
DllFree(UserLink)
DialogAddGroupBox("Dlg1"; "Gb1"; 91; 4; 70; 38; "Orientation")
DialogAddRadioButton("Dlg1"; "Rb1"; 101; 14; 50; 11; "&Portrait";Port)
DialogAddRadioButton("Dlg1"; "Rb2"; 101; 26; 50; 11; "&Landscape";Land)
DialogAddPushButton("Dlg1"; "Pb1"; 107; 49; 38; 13; 0; "P&review")
Label(MainDialog@)
Display(On!)
DialogDisplay("Dlg1"; "Lb1")
DialogResult:=MacroDialogResult
Display(Off!)

If (DialogResult =2)
Go(End@)

EndIf

For(Count; 1; Border[Count; 1]<>Choice; Count+1)
EndFor

If (Land = 1)
PorL:=2 // Landscape

Else
PorL:=3 // Portrait

EndIf

FileExists(FileExist; Path+Border[Count; PorL])
If(Not FileExist)

MessageBox(MsgBoxResult; "Graphic Not Found"; "^0 does not exist.  Check your 
graphics directory in preferences, or press Retry to select a different directory."; 
IconExclamation! | RetryCancel! | Beep! | HasParameters!; Path+Border[Count; PorL])
If(MsgBoxResult=4)

Call(CheckDir@)
EndIf
Go(MainDialog@)

EndIf
If(DialogResult="Pb1")

Call(ViewDlg@)
Go(MainDialog@)



EndIf

BoxCreate(FigureBox!)
BoxAttachTo(Page!)
BoxPageOffset(No!)
BoxHorizontalPosition(0.0")
BoxVerticalPosition(0.0")
BoxHeight(HeightOfBox)
BoxWidth(WidthOFBox)
BoxBorder(0)
BorderSetSpacing(Yes!)
BorderOutsideSpacing(0.0"; 0.0"; 0.0"; 0.0")
BoxTextFlow(NoWrapping!)
BoxOverlap(Yes!)
BoxContentType(Image!)
BoxImageRetrieve(MakeInternal!; Path+Border[Count; PorL])
BoxEnd(Save!)
CloseGraphicsControlBar()

Label(End@)
If(DlgDefined)

DialogDestroy("Dlg1")
EndIf
Quit // End Macro

//***************************************************************
//* ROUTINE:  ViewDlg@
//* INPUT VARIABLES:  Path; Border[Count; PorL]; Land
//* OUTPUT VARIABLES:  None
//* DESCRIPTION:  Displays a viewer for the highlighted file.
//***************************************************************
Label(ViewDlg@)
If(Land=1)

DlgHeight:=150
DlgWidth:=175

Else
DlgHeight:=176
DlgWidth:=116

EndIf
DialogDefine("Dlg2"; 50; 50; DlgWidth+20; DlgHeight+50; 1+16; "Preview Border")
DialogAddViewer("Dlg2"; "V1"; 7; 8; DlgWidth; DlgHeight; Path+Border[Count; PorL])
DialogDisplay("Dlg2"; 1)
DialogDestroy("Dlg2")
Return
//***************************************************************



//***************************************************************
//* ROUTINE:  Definitions@
//* INPUT VARIABLES:  None
//* OUTPUT VARIABLES:  Border[]
//* DESCRIPTION:  Defines the array for the page borders.
//***************************************************************
Label(Definitions@)
MaxBorder:=11 // This must be increased if more are added
Declare(Border[MaxBorder; 3])

// Format for border definitions:     
Border[1; 1]:="Ornate" // Descriptive Name
Border[1; 2]:="bord01l.wpg" // Landscape File
Border[1; 3]:="bord01p.wpg"// Portrait File

Border[2; 1]:="Modern"
Border[2; 2]:="bord02l.wpg"
Border[2; 3]:="bord02p.wpg"

Border[3; 1]:="Classic"
Border[3; 2]:="bord03l.wpg"
Border[3; 3]:="bord03p.wpg"

Border[4; 1]:="Fancy"
Border[4; 2]:="bord04l.wpg"
Border[4; 3]:="bord04p.wpg"

Border[5; 1]:="Safety Border"
Border[5; 2]:="bord05l.wpg"
Border[5; 3]:="bord05p.wpg"

Border[6; 1]:="Square Corners"
Border[6; 2]:="bord06l.wpg"
Border[6; 3]:="bord06p.wpg"

Border[7; 1]:="Pencils"
Border[7; 2]:="bord07l.wpg"
Border[7; 3]:="bord07p.wpg"

Border[8; 1]:="Starburst"
Border[8; 2]:="bord08l.wpg"
Border[8; 3]:="bord08p.wpg"

Border[9; 1]:="Confetti"
Border[9; 2]:="bord11l.wpg"
Border[9; 3]:="bord11p.wpg"



Border[10; 1]:="Fleur-de-Lis"
Border[10; 2]:="bord12l.wpg"
Border[10; 3]:="bord12p.wpg"

Border[11; 1]:="Snapshot"
Border[11; 2]:="bord13l.wpg"
Border[11; 3]:="bord13p.wpg"
Return
//***************************************************************

//***************************************************************
//* ROUTINE:  CheckDir@
//* INPUT VARIABLES:  Path
//* OUTPUT VARIABLES:  Path
//* DESCRIPTION:  This will create a dialog to prompt for a directory.
//***************************************************************
Label(CheckDir@)
DialogDefine("PathDlg"; 50; 50; 172; 90; 1+2+16; "Graphic Directory")
DialogAddFilenameBox("PathDlg"; "Fb1"; 8; 8; 150; 14; 1; Path; Path; "*.wpg")
DialogAddText("PathDlg"; "T1"; 8; 28; 150; 20; 1; "Please select the directory that contains the 
Page Border graphics files.")
Label(Redisplay@)
Display(On!)
DialogDisplay("PathDlg"; "Fb1")
Display(Off!)
If(MacroDialogResult=2)

DialogDestroy("PathDlg")
Go(End@)

EndIf
If(Path="")

Go(Redisplay@)
EndIf
DialogDestroy("PathDlg")
If(SubStr(Path; StrLen(Path); 1)<>"\")

Path:=Path+"\"
EndIf
Return
//***************************************************************


